Лабораторна робота №2
Простий пристрій моніторингу температури

Мета роботи: Оволодіти основними принципами роботи з Arduino та підключенням різних тепературних датчиків до плати, ознайомитись з різновидами функцій та команд для створення програми для зняття данних про температуру.

1. Теоретичні відомості
1.1 Послідовний інтерфейс UART 
 В перекладі це універсальний асинхронний приймач. Дані UART передаються послідовним кодом в наступному форматі.

Рис 1.1 Схема формату передачі данних

Кожен біт передається за рівні проміжки часу. Час передачі одного біта визначається швидкістю передачі. Швидкість передачі вказується в бодах (біт в секунду). Крім бітів даних інтерфейс UART вставляє в потік біти синхронізації: стартовий і стоповий. Таким чином, для передачі байта інформації потрібно 10 бітів, а не 8. Похибка тимчасових інтервалів передачі бітів повинна бути не більше 5% (рекомендується не більше 1,5%).

Існують варіанти з різною кількістю бітів даних, бітів синхронізації, може бути доданий біт контролю парності і т.п. Але ці формати використовуються рідко. Головне запам'ятати:

· в неактивному режимі вихід UART знаходиться в високому стані;

· передача байта починається зі стартового біта (низького рівня);

· передача байта закінчується стоповим бітом (високого рівня);

· дані передаються молодшим бітом вперед;

· для передачі байта потрібно 10 бітів;

Час передачі одного байта розраховується виходячи з швидкості передачі і кількості бітів (10).
Часто використовуються такі стандартні швидкості передачі інтерфейсу UART.

Обмін інформацією через UART відбувається в дуплексному режимі, тобто передача даних може відбуватися одночасно з прийомом. Для цього в інтерфейсі UART є два сигнали:

TX - вихід для передачі даних;
RX - вхід для прийому даних.

При з'єднанні двох UART пристроїв вихід TX одного пристрою з'єднується з входом RX іншого. А сигнал TX другого UART підключається до входу RX першого.

1.2 Протокол 1-Wire

Протокол 1-Wire - це протокол, який використовується для управління пристроями, які виробляються компанією Dallas Semiconductor (нині Maxim). Хоча 1-Wire є торговою маркою Dallas, програмісти, які використовують драйвери 1-Wire, не зобов'язані робити за це ніяких виплат.

Мережа 1 Wire, яку Даллас називає MicroLan (торгова марка), складається з одного ведучого пристрою, до якого за допомогою єдиної лінії передачі даних з'єднання з одним або декілька ведених пристроїв. Ця лінія, крім іншого, можна використовувати для електроживлення ведених пристрої (ситуація, коли пристрої підживлюються через шину 1-Wire, називають «паразитне живлення»).

Серед інших пристроїв, що працюють через протокол 1-Wire, особливо популярні температурні датчики - вони недорогі, прості у використанні і дозволяють безпосередньо зчитувати калібровані цифрові температурні дані. Крім того, вони терпимі до довгих проводах, якими часто доводиться користуватися при створенні ланцюга з Arduino. Один із прикладів нижче демонструє, як працювати з 1-Wire на прикладі цифрового термометра DS18S20. Багато Чіпи 1 Wire можуть працювати і через паразитне, і через нормальне живлення.

1.3 USB
Універсальна послідовна шина, призначила для з'єднанні періферійніе Пристрій обчіслювальної техніки.
USB роз'єм типу А — найпоширеніший і найвідоміший із нинішніх. Більшість пристроїв, що підключаються через USB, мають саме його. Комп'ютерна миша, USB флеш-накопичувач, клавіатура, фото- й відеокамера та багато інших речей оснащені USB типу A, який бере свій початок ще з 90-х. Одна з найголовніших переваг цього порту — надійність. Він може пережити досить велику кількість підключень, не розвалюється й дійсно заслужив стати найпоширенішим засобом підключення всього, чого тільки можна. Однак для портативних пристроїв він не підходить, тому що має досить великі габарити, що врешті-решт призвело до появи модифікацій із меншими розмірами — Mini та Micro.

USB роз'єм типу В найчастіше використовується для приєднання до комп'ютера принтерів і сканерів, зрідка — інших пристроїв.

Розроблені також типи Mini-AB та Micro-AB для з'єднання через конектори відповідного розміру як типу А, так і типу В.

В даний час існує декілька видів роз'ємів USB, які бувають трьох версій - USB v1.1, USB v2.0 і USB v3.0. Принципова різниця між ними , тільки в швидкості передачі данних яка в  USB v3.0 найшвидша 500 Мбайт/сек , USB v2.0 48 Мбайт/сек і USB v1.1 12 Мбіт/сек .

1.3 Датчик температури DS18B20

DS18B20 має різні форм-фактори.Доступно три варіанти: 8-Pin SO (150 mils), 8-Pin μSOP, і 3-Pin TO-92. Серфінг по eBay або Aliexpress показує, що китайці пропонують версію TO-92 у вологозахищеному корпусі. Тобто, можна сміливо занурювати подібне диво в воду, використовувати під дощем і т.д. і т.п. Ці сенсори виготовляються з трьома вихідними контактами (чорний - GND, червоний - Vdd і білий - Data).

Рис 1.2 Схема датчика температури

DS18B20 зручний у використанні. Живити його можна через контакт data (в такому випадку використовується всього два контакти з трьох для підключення). Сенсор працює в діапазоні напруг від 3.0 В до 5.5 В і вимірює температуру в діапазоні від -55 ° C до + 125 ° C (від -67 ° F до + 257 ° F) з точністю ± 0.5 ° C (від -10 ° C до + 85 ° C).

Також можна підключити паралельно до 127 датчиків і прочитувати свідчення температури з кожного окремо. Не зовсім зрозуміло, в якому проекті подібне може знадобиться, але підключити два сенсора і контролювати температуру в холодильнику і морозильній камері можна. При цьому останеться вільними купу пинов на Arduino.


2. Завдання до лабораторної роботи

2.1 Підключіть датчик до плати Arduino згідно зі схемою, яка зображена на рисунку 2.1.

Рисунок 2.1 Схема підлючення датчика температури

2.2 Запустіть на комп’ютері середовища розробки Arduino.
2.3 Підключіть плати Arduino через USB до ПК.
2.4 Написати код програми за таким алгоритмом.

2.5 Натисніть кнопку Verify и переконайтесь, що у нижній частині вікна з’явився надпис Done Compiling. Це означає, що у написаній програмі не знайдено помилок.
2.6 Виберіть у Tools->Board ваш тип плати. Перевірте, чи правильно обрано USB-порт в Tools->Serial port. Після натисніть на кнопку Upload.
Якщо внизу з’явився надпис “Done uploading” – процес запису прошов успішно.


3. Зміст звіту

3.1. Тема та мета роботи.
3.2. Схема електрична-принципова плати Arduino. 
3.3. Код програми.
3.4. Висновки з роботи.

4. Контрольні запитання

1. Що таке послідовний інтерфейс UART ?
2. Як відбувається передача данних в UART?
3. Що таке протокол 1-Wire і навіщо він потрібний ?
4. Скільки існує  видів USB роз'ємів і чим вони відрізняються ? 
5. Скільки контактів має датчик температури і що кожний з них означає?

Скетч
//підключити бібліотеки OneWire і DallasTemperature
#include <OneWire.h>
#include <DallasTemperature.h>

// вказати  до якого порту ми підключили контакт по передачі даннихв нашому випадку до 2
#define ONE_WIRE_BUS 2
#define TEMPERATURE_PRECISION 9
//налаштування для oneWire для спілкування з будь-якими пристроями-Wire
OneWire oneWire(ONE_WIRE_BUS);
DallasTemperature sensors(&oneWire);
// масиви для зберігання адрес пристроїв
DeviceAddress insideThermometer, outsideThermometer;
void setup(void)
{
  // почати  послідовного порту
  Serial.begin(9600);
  Serial.println("Dallas Temperature IC Control Library Demo");

  // Запуск бібліотеки
  sensors.begin();

  // знайти пристрої на шині
  Serial.print("Locating devices...");
  Serial.print("Found ");
  Serial.print(sensors.getDeviceCount(), DEC);
  Serial.println(" devices.");

  // вказуємо вимоги до паразитного живлення
  Serial.print("Parasite power is: "); 
  if (sensors.isParasitePowerMode()) Serial.println("ON");
  else Serial.println("OFF");

   // алгоритм пошуку виглядає так . Повертає 1, якщо нова адреса була знайдена
   // нуль може означати, що шина замкнута, немає пристроїв,
   // або  вже витягнуто всі з них
   // перевірка CRC, щоб переконатися, що ми не отримаємо сміття
   // привласнює першу адресу, вказану в insideThermometer
   // якщо Serial.println ( "Не вдалося знайти адресу для insideThermometer") (oneWire.search (insideThermometer)!);
   // привласнює секундної адресу, вказану в outsideThermometer
   // якщо Serial.println ( "Не вдалося знайти адресу для outsideThermometer") (oneWire.search (outsideThermometer)!);

  if (!sensors.getAddress(insideThermometer, 0)) Serial.println("Unable to find address for Device 0"); 
  if (!sensors.getAddress(outsideThermometer, 1)) Serial.println("Unable to find address for Device 1");
	
   // показати адреси які ми знайшли на шині
Serial.print("Device 0 Address: ");
  printAddress(insideThermometer);
  Serial.println();

  Serial.print("Device 1 Address: ");
  printAddress(outsideThermometer);
  Serial.println();

// встановити дозвіл на  9 біт
sensors.setResolution(insideThermometer, TEMPERATURE_PRECISION);
  sensors.setResolution(outsideThermometer, TEMPERATURE_PRECISION);

  Serial.print("Device 0 Resolution: ");
  Serial.print(sensors.getResolution(insideThermometer), DEC); 
  Serial.println();

  Serial.print("Device 1 Resolution: ");
  Serial.print(sensors.getResolution(outsideThermometer), DEC); 
  Serial.println();
}

// Функція для друку адресу пристрою
void printAddress(DeviceAddress deviceAddress)
{
  for (uint8_t i = 0; i < 8; i++)
  {
   
    if (deviceAddress[i] < 16) Serial.print("0");
    Serial.print(deviceAddress[i], HEX);
  }
}

// Функція для друку температури для пристрою
void printTemperature(DeviceAddress deviceAddress)
{
  float tempC = sensors.getTempC(deviceAddress);
  Serial.print("Temp C: ");
  Serial.print(tempC);
  Serial.print(" Temp F: ");
  Serial.print(DallasTemperature::toFahrenheit(tempC));
}

// Функція для друку дозволу пристрою
void printResolution(DeviceAddress deviceAddress)
{
  Serial.print("Resolution: ");
  Serial.print(sensors.getResolution(deviceAddress));
  Serial.println();    
}

// головна функція для друку інформації про пристрій
void printData(DeviceAddress deviceAddress)
{
  Serial.print("Device Address: ");
  printAddress(deviceAddress);
  Serial.print(" ");
  printTemperature(deviceAddress);
  Serial.println();
}

void loop(void)
{ 
 // sensors.requestTemperatures виклик для видачі глобальної температури
   // запит на всі пристрої на шині

  Serial.print("Requesting temperatures...");
  sensors.requestTemperatures();
  Serial.println("DONE");

 // друк інформації про пристрій
  printData(insideThermometer);
  printData(outsideThermometer);
}

